# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=python-ziamath
pkgver=0.8
pkgrel=1
pkgdesc="Render MathML and LaTeX Math to SVG without Latex installation"
url="https://ziamath.readthedocs.io/"
license=('MIT')
arch=('any')
depends=('python-ziafont' 'python-latex2mathml')
makedepends=('python-build' 'python-installer' 'python-setuptools' 'python-wheel')
source=("$pkgname-$pkgver.tar.gz::https://github.com/cdelker/ziamath/archive/$pkgver.tar.gz")
sha512sums=('6f11203407c84b160c449f758bdd865c1247298c8f4f9e94b5a38fe1d39ede80fe0522316f686f21423cfc93b6f66b0697002992544734da9ed8d851b937a5c2')

build() {
  cd ziamath-$pkgver
  python -m build -nw
}

package() {
  cd ziamath-$pkgver
  python -m installer -d "$pkgdir" dist/*.whl
  install -Dm644 LICENSE.txt -t "$pkgdir"/usr/share/licenses/$pkgname/
}
